
1. Change

	Type/Function name
	IMS_DialogInfo_Type, f_IMS_DialogInfo_Init, f_IMS_PTC_ImsInfo_GetURI

	Reason for change
	In a dialog the same URIs shall be used in From and To headers as in the initial message of the dialog

	Summary of change
	RemoteURI and LocalURI changed in IMS_DialogInfo_Type to SipUrl (instead of charstring); f_IMS_PTC_ImsInfo_GetURI renamed to f_IMS_PTC_ImsInfo_GetUriString (all occurencies)

	TTCN module
	IMS_Component


	  type record IMS_DialogInfo_Type {                     /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, POS)

                                                           @sic R5s140092 change 1.5: RemoteURI, RemoteCseq, LocalURI, LocalCseq stored in this structure sic@ */

    boolean             MobileOriginated  optional,     /* needed e.g. to get the right route-set for re-INVITE */

    charstring          RemoteTag         optional,     /* acc. to RFC 3261 when creating a dialog in first request there is a From tag but no To tag;

                                                           the To tag is chosen by the other end and sent in a response; in all further messages within the dialog the same From and To tags shall be used

                                                           (=> especially in request messages (e.g. BYE) it may be necessary to use this stored value whereas in responses the values can be used as sent in the request);

                                                           NOTE: in general outside a dialog (e.g. registration) requests get a new From tag and the corresponding reponses contain a new To tag */ 
    SipUrl              RemoteURI         optional,     /* URI of the remote side in a dialog created by INVITE */

    integer             RemoteCseq        optional,     /* @sic R5s140235: changed to optional sic@ */

    charstring          LocalTag          optional, 
    SipUrl              LocalURI          optional,     /* URI of the UE in a dialog created by INVITE */

    integer             LocalCseq,

    charstring          CallId            optional,

    charstring          SessionId         optional,

    SipUrl              ContactUrl        optional,     /* Contact Url provided by the UE for this dialog (especially in case GRUU this different to the (non-GRUU) URI used for registration); @sic R5s150908 sic@ */

    charstring          SdpSessionIdTX,                 /* @sic R5-150707 sic@ */

    integer             SdpSessionVersionTX,            /* current SDP session version for the DL   @sic R5-150707 sic@ */

    SDP_Message         SdpMessagePrevRX  optional      /* previous SDP message sent by the UE      @sic R5-150702 sic@ */

  };


	  function f_IMS_DialogInfo_Init(IMS_DialogInfo_Type p_DialogInfo,

                                 IMS_PTC_CseqType_Type p_CseqType,

                                 MessageHeader p_MessageHeader_InviteOrReferOrSubscribe) return IMS_DialogInfo_Type

  { 
    var SipUrl v_FromURI := f_Addr_Union_GetSipUrl(p_MessageHeader_InviteOrReferOrSubscribe.fromField.addressField);

    var SipUrl v_ToURI   := f_Addr_Union_GetSipUrl(p_MessageHeader_InviteOrReferOrSubscribe.toField.addressField);


    var charstring v_FromTag := f_MessageHeader_GetFromTag(p_MessageHeader_InviteOrReferOrSubscribe);

    var integer v_CseqValue := p_MessageHeader_InviteOrReferOrSubscribe.cSeq.seqNumber;

    var IMS_DialogInfo_Type v_Dialog := p_DialogInfo;

    select (p_CseqType) {

      case (dialogLocal) {                 /* MT establishment of the dialog */

        v_Dialog.LocalURI  := v_FromURI;   /* @sic R5s140121 change 2.2: needs to be v_FromURI instead of v_ToURI sic@ */

        v_Dialog.RemoteURI := v_ToURI;     /* @sic R5s140121 change 2.2: needs to be v_ToURI instead of v_FromURI sic@ */

        v_Dialog.MobileOriginated := false;

        v_Dialog.LocalTag := v_FromTag;

        v_Dialog.RemoteTag := omit;        /* @sic R5s130497 change 1 - MCC160 implementation sic@ */

        v_Dialog.LocalCseq := v_CseqValue;

      }

      case (dialogRemote) {                /* MO establishment of the dialog */

        v_Dialog.LocalURI  := v_ToURI;     /* @sic R5s140121 change 2.2: needs to be v_ToURI instead of v_FromURI sic@ */

        v_Dialog.RemoteURI := v_FromURI;   /* @sic R5s140121 change 2.2: needs to be v_FromURI instead of v_ToURI sic@ */

        v_Dialog.MobileOriginated := true;

        v_Dialog.LocalTag := omit;         /* @sic R5s130497 change 1 - MCC160 implementation sic@ */

        v_Dialog.RemoteTag := v_FromTag;

        v_Dialog.RemoteCseq := v_CseqValue;

        v_Dialog.ContactUrl := f_MessageHeader_GetContactSipUrl(p_MessageHeader_InviteOrReferOrSubscribe);

      }

      case else {

        FatalError (__FILE__, __LINE__, "");

      }

    }

    v_Dialog.CallId := p_MessageHeader_InviteOrReferOrSubscribe.callId.callid;

    if (ispresent(p_MessageHeader_InviteOrReferOrSubscribe.sessionId)) {

      v_Dialog.SessionId := p_MessageHeader_InviteOrReferOrSubscribe.sessionId.sessid;

    }

    return v_Dialog;

  }


	  function f_IMS_PTC_ImsInfo_DialogGetSipUri(IMS_UriInfo_UriType_Type p_IdType,

                                             template (omit) IMS_PTC_DialogIndex_Type p_DialogIndex := omit) runs on IMS_PTC return template (omit) SipUrl
  {

    var IMS_DialogInfo_Type v_Dialog := fl_IMS_PTC_ImsInfo_GetDialog(p_DialogIndex);

    var template (omit) SipUrl v_SipUri := omit;

    select (p_IdType) {

      case (dialogLocalId) {

        v_SipUri := v_Dialog.LocalURI;

      }

      case (dialogRemoteId) {

        v_SipUri := v_Dialog.RemoteURI;

      }

    }

    return v_SipUri;

  }


	  function  f_IMS_PTC_ImsInfo_GetUriString(IMS_UriInfo_UriType_Type p_IdType,

                                    template (omit) IMS_PTC_DialogIndex_Type p_DialogIndex := omit) runs on IMS_PTC return charstring

  {

 
    var template (omit) SipUrl v_SipUrl;
    select (p_IdType) {

      case (firstId, defaultId, emergencyId) {

        return vc_IMS_Global.ImsInfo.IMPUList[0];

      }

      case (secondId) {

        return vc_IMS_Global.ImsInfo.IMPUList[1];

      }

      case (thirdId) {

        return vc_IMS_Global.ImsInfo.IMPUList[2];

      }

      case (registrationId, gruu) {     /* @sic R5-153760: support of gruu; same URI as used by the UE in the To header of the REGISTER message sic@ */

        if (ispresent(vc_IMS_Global.ImsInfo.Registration.Uri)) {

          return vc_IMS_Global.ImsInfo.Registration.Uri;

        }

      }
      case (dialogLocalId, dialogRemoteId) {

        v_SipUrl := f_IMS_PTC_ImsInfo_DialogGetSipUri(p_IdType, p_DialogIndex); 
        if (isvalue(v_SipUrl)) {

          return f_SIP_SipUrl2Str(valueof(v_SipUrl));

        }

      }










      case (temporaryId) {

        return f_IMS_TemporaryPublicUserIdentity(px_IMSI_Def);

      }

    }

    FatalError(__FILE__, __LINE__, "Invalid URI or URI not initialised");

    return "";

  }


2. Change

	Function name
	f_IMS_PTC_ImsInfo_GetSipUriTX, f_IMS_PTC_ImsInfo_GetSipUriRX

	Reason for change
	In case of an existing dialog the same URI shall be used in the From/To headers of subsequent messages as in the initial message creating the dialog

	Summary of change
	In case of existing dialog the stored SipUrl values are used (including URI parameters); formal parameter p_Port removed from f_IMS_PTC_ImsInfo_GetSipUriTX as never a value is handed over ("-" removed for calls in f_IMS_RequestInDialog_FromHeaderTX and f_IMS_RequestInDialog_ToHeaderTX)

	TTCN module
	IMS_Component


	  function f_IMS_PTC_ImsInfo_GetSipUriTX(IMS_UriInfo_UriType_Type p_IdType,


                                         template (omit) IMS_PTC_DialogIndex_Type p_DialogIndex := omit) runs on IMS_PTC return template (value) SipUrl

  { /* @sic R5s150382 change 1: p_DialogIndex sic@ */ 
    var template (omit) SipUrl v_DialogSipUrl := f_IMS_PTC_ImsInfo_DialogGetSipUri(p_IdType, p_DialogIndex);

    var template (value) SipUrl v_SipUrl;

    if (isvalue(v_DialogSipUrl)) {

      v_SipUrl := valueof(v_DialogSipUrl);

    } else {

      v_SipUrl := f_SIP_BuildSipUri_TX(f_IMS_PTC_ImsInfo_GetUriString(p_IdType, p_DialogIndex));

    }

    return v_SipUrl;

  }


	  function f_IMS_PTC_ImsInfo_GetSipUriRX(IMS_UriInfo_UriType_Type p_IdType,

                                         template integer p_Port := omit,

                                         template (omit) IMS_PTC_DialogIndex_Type p_DialogIndex := omit) runs on IMS_PTC return template (present) SipUrl

  { /* @sic R5s150382 change 1: p_DialogIndex sic@ */ 
    var template (omit) SipUrl v_DialogSipUrl := f_IMS_PTC_ImsInfo_DialogGetSipUri(p_IdType, p_DialogIndex); 
    var template (present) SipUrl v_SipUrl;

    if (isvalue(v_DialogSipUrl)) {

      v_SipUrl := valueof(v_DialogSipUrl);

    } else {

      v_SipUrl := f_SIP_BuildSipUri_RX(f_IMS_PTC_ImsInfo_GetUriString(p_IdType, p_DialogIndex), p_Port);

    }

    return v_SipUrl;

  }


3. Change

	Function name
	f_IMS_RegisterRequest_MessageHeaderRX_Subsequent, f_IMS_RegisterRequest_MessageHeaderRX_Derregister

	Reason for change
	Improvement: Use of f_IMS_PTC_ImsInfo_GetSipUriRX in f_IMS_RegisterRequest_MessageHeaderRX_Subsequent, f_IMS_RegisterRequest_MessageHeaderRX_Derregister

	Summary of change
	"f_SIP_BuildSipUri_RX(f_IMS_PTC_ImsInfo_GetUriString(registrationId))" replaced by "f_IMS_PTC_ImsInfo_GetSipUriRX(registrationId)"

	TTCN module
	IMS_Procedures_Registration


	  function f_IMS_RegisterRequest_MessageHeaderRX_Subsequent(integer p_NonceCount,

                                                            InternetProtocol_Type p_Protocol) runs on IMS_PTC return template (present) MessageHeader

  { /* p_NonceCount  ..  Acc. to A.1.1 "counter to indicate how many times UE has sent the same value of nonce within successive REGISTERs, initial value shall be 1"

     *                   => during initial registration "1" is used per default; for each re-authentication (=> new nonce) the nonce-counter is reset to "1"

     * @sic R5-142727: new parameter p_Protocol sic@ */

    var template (present) MessageHeader v_MessageHeader_Register;

    var template (present) SipUrl v_SipUrlPublicUserId;

    var charstring v_Nonce := f_IMS_PTC_Security_GetNonce();

    var charstring v_NonceCountStr := "0000000" & int2str(p_NonceCount);     // assuming p_NonceCount < 9

    var IMS_ProtectedPorts_Type v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

    var integer v_CseqValue := f_IMS_PTC_ImsInfo_CseqIncr(register);

    var boolean v_IsGiba := false;

    var template integer v_ViaPort := *;

    if (p_Protocol == udp) {      // @sic R5-145727 sic@

      v_ViaPort := v_Protected.Port_us;

    }

    v_SipUrlPublicUserId := f_IMS_PTC_ImsInfo_GetSipUriRX(registrationId);
    /* public user identity is checked against what the UE has sent in initial REGISTER

                                                                                                     (in case the UE does not have an ISIM this is the temporary user id)

                                                                                                     => we don't need to consider A7 (emergency registration) */

    v_MessageHeader_Register := fl_IMS_RegisterRequest_MessageHeaderRX_Base(v_IsGiba, v_CseqValue);

    v_MessageHeader_Register.route       := omit;                                                // @sic R5-134659 sic@

    v_MessageHeader_Register.via         := cr_ViaDef(-, cr_HostPort(?, v_ViaPort));             // @sic R5s120727 change 7; R5-145727 sic@

    v_MessageHeader_Register.fromField   := cr_FromWithTag(v_SipUrlPublicUserId);

    v_MessageHeader_Register.toField     := cr_ToDef(v_SipUrlPublicUserId);

    v_MessageHeader_Register.contact     := cr_Contact(cr_SipUri_HostPort(?, v_Protected.Port_us));   // checked in f_IMS_A_1_1_MessageHeader_CommonChecks; @sic R5s120727 change 7 sic@

    v_MessageHeader_Register.pAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString()); // cell id only if applicable

    v_MessageHeader_Register.securityVerify     := f_IMS_PTC_BuildSecurityVerifyHeader();

    v_MessageHeader_Register.authorization      := cr_AuthorizationDef(f_IMS_PTC_ImsInfo_GetPrivateUserId(), f_IMS_PTC_ImsInfo_GetHomeDomainName(), v_Nonce, v_NonceCountStr);

    return v_MessageHeader_Register;

  }


	  function f_IMS_RegisterRequest_MessageHeaderRX_Derregister() runs on IMS_PTC return template (present) MessageHeader

  {

    var boolean v_CondA3 := f_IMS_PTC_SecurityScheme_IsGiba();  // condition A3 acc. to 34.229-1 Annex A.1.1 (REGISTER for the case UE supports GIBA)

    var template (present) MessageHeader v_MessageHeader_Register; 
    var template (present) SipUrl v_SipUrlPublicUserId := f_IMS_PTC_ImsInfo_GetSipUriRX(registrationId);
    /* public user identity is checked against what the UE has sent in initial REGISTER

                                                                                                                              NOTE: there is no indication in 34.229-1 to check this */

    var template (present) integer v_CseqValue := f_IMS_PTC_ImsInfo_CseqIncr(register);

    var IMS_ProtectedPorts_Type v_Protected;

    var charstring v_Nonce;

    v_MessageHeader_Register := fl_IMS_RegisterRequest_MessageHeaderRX_Base(v_CondA3, v_CseqValue);

    // common parameters for A2 and A3:

    v_MessageHeader_Register.fromField   := cr_FromWithTag(v_SipUrlPublicUserId);

    v_MessageHeader_Register.toField     := cr_ToDef(v_SipUrlPublicUserId);

    v_MessageHeader_Register.supported   := *;    // Supported: header may be missing or it may contain any value

    v_MessageHeader_Register.route       := omit; // @sic R5-134659 sic@

    if (v_CondA3) {

      v_MessageHeader_Register.via       := cr_ViaDef;

      v_MessageHeader_Register.contact   := (cr_Contact_Wildcard, cr_Contact(cr_SipUri_HostPort(?, *)));

    }

    else {

      v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

      v_Nonce := f_IMS_PTC_Security_GetNonce();

      v_MessageHeader_Register.via       := cr_ViaDef(-, cr_HostPort(?, v_Protected.Port_us));                             // address will be checked in f_IMS_A_1_1_MessageHeader_CommonChecks (@sic R5s120727 sic@)

      v_MessageHeader_Register.contact   := (cr_Contact_Wildcard, cr_Contact(cr_SipUri_HostPort(?, v_Protected.Port_us))); // address will be checked in f_IMS_A_1_1_MessageHeader_CommonChecks (@sic R5s120727 sic@)

      v_MessageHeader_Register.pAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString()); // cell id only if applicable

      v_MessageHeader_Register.securityVerify     := f_IMS_PTC_BuildSecurityVerifyHeader();

      v_MessageHeader_Register.authorization      := cr_AuthorizationDef(f_IMS_PTC_ImsInfo_GetPrivateUserId(), f_IMS_PTC_ImsInfo_GetHomeDomainName(), v_Nonce, ?);

    }

    return v_MessageHeader_Register;

  }


